LABORATORY 6: Solutions of second order PDE with two variables

Initialization

```
> restart:
> with(DEtools):
> with(PDEtools):
> with(plots):
Warning, the name changecoords has been redefined

>
```

The Cauchy problem for the infinite vibrating string equation

$$\left(\frac{\partial^2}{\partial t^2}\mathrm{u}(t,x)\right) - a^2\left(\frac{\partial^2}{\partial x^2}\mathrm{u}(t,x)\right) = \mathrm{f}(t,x)\,,\ x \in IR\,,\ t \in (0,T)$$

$$\mathrm{u}(0,x) = u_0(x)$$

$$\frac{\partial}{\partial t}\mathrm{u}(t,x)\ (0,x) = u_1(x)$$

Solution is given by the D'Alembert formula:

$$\mathrm{u}(t,x) = \frac{u_0(x-a\,t)+u_0(x+a\,t)}{2} + \frac{\int_{x-a\,t}^{x+a\,t} u_1(s)\,ds}{2\,a} + \frac{\int_0^t \int_{x-a\,(t-\tau)}^{x+a\,(t-\tau)} \mathrm{f}(\tau,s)\,ds\,d\tau}{2\,a}$$

Let's find the solution in the case of $u_0 = \cos(x^2)$, $u_1(x) = 2\sin(2\,x)$, $\mathrm{f}(t,x) = 2\,t\,x$, $a = 1$

```
> PDE:=diff(u(t,x),t$2)-a^2*diff(u(t,x),x$2)=f(t,x);
```

$$PDE := \left(\frac{\partial^2}{\partial t^2}\mathrm{u}(t,x)\right) - a^2\left(\frac{\partial^2}{\partial x^2}\mathrm{u}(t,x)\right) = \mathrm{f}(t,x)$$

```
> a:=1;
```

$$a := 1$$

```
> u0:=x->cos(x^2);
```

$$u0 := x \to \cos(x^2)$$

```
> u1:=x->2*sin(2*x);
```

$$u1 := x \to 2\sin(2\,x)$$

```
> f:=(t,x)->2*t*x;
```

$$f := (t,x) \to 2\,x\,t$$

```
> u_sol:=(t,x)->(u0(x-a*t)+u0(x+a*t))/2+int(u1(s),s = x-a*t ..
x+a*t)/(2*a)+int(int(f(tau,s),s = x-a*(t-tau) .. x+a*(t-tau)),tau = 0 ..
t)/(2*a);
```

$$u\_sol := (t,x) \to \frac{1}{2}\,\mathrm{u0}(x - a\,t) + \frac{1}{2}\,\mathrm{u0}(x + a\,t) + \frac{1}{2}\left(\frac{1}{a}\int_{x-a\,t}^{x+a\,t} \mathrm{u1}(s)\,ds\right) + \frac{1}{2}\left(\frac{1}{a}\int_{0}^{t}\int_{x-a\,(t-\tau)}^{x+a\,(t-\tau)} \mathrm{f}(\tau, s)\,ds\,d\tau\right)$$

```
> u_sol(t,x);
```

$$\frac{1}{2}\cos((x-t)^2) + \frac{1}{2}\cos((x+t)^2) - \frac{1}{2}\cos(2\,x + 2\,t) + \frac{1}{2}\cos(2\,x - 2\,t) - \frac{2\,x\,t^3}{3} + \frac{((x+t)^2 - (x-t)^2)\,t^2}{4}$$

Let's check if, indeed, this function is solution of PDE

```
> pdetest(u(t,x)=u_sol(t,x),PDE);
```

$$0$$

and initial conditions:

```
> u_sol(0,x);
```

$$\cos(x^2)$$

```
> D[1](u_sol)(0,x);
```

$$2\sin(2\,x)$$

We can plot the solution in 3d with respect to variable t and x

```
> plot3d(u_sol(t,x),t=0..2,x=-2..2,axes=NORMAL);
```



or we can see the evolution of the solution in time using animate command

```
> animate(u_sol(t,x),x=-2..2,t=0..2);
```



The mixed problem for the vibrating string equation

$$\left(\frac{\partial^2}{\partial t^2}\mathrm{u}(t,x)\right) - a^2\left(\frac{\partial^2}{\partial x^2}\mathrm{u}(t,x)\right) = \mathrm{f}(t,x)\ ,\ \ x \in (0,l)\ ,\ \ t \in (0,T)$$

$$\mathrm{u}(0,x) = u_0(x)$$

$$\frac{\partial}{\partial t}\mathrm{u}(t,x)\ \ (0,x) = u_1(x)$$

$$\mathrm{u}(t,0) = \mathrm{u}(t,l) = 0$$

The solution has the following form:

$$\mathrm{u}(t,x) = \mathrm{v}(t,x) + \mathrm{w}(t,x)$$

where

$$\mathrm{v}(t,x) = \sum_{k=1}^{\infty}\left(a_k\cos\left(\frac{a\,k\,\pi\,t}{l}\right) + b_k\sin\left(\frac{a\,k\,\pi\,t}{l}\right)\right)\sin\left(\frac{k\,\pi\,x}{l}\right)$$

$$a_k = \frac{2\int_0^l u_0(x)\sin\left(\frac{k\,\pi\,x}{l}\right)dx}{l}$$

$$b_k = \frac{2\int_0^l u_1(x)\sin\left(\frac{k\,\pi\,x}{l}\right)dx}{a\,k\,\pi}$$

$$\mathrm{w}(t,x) = \sum_{k=1}^{\infty}\phi_k(t)\sin\left(\frac{k\,\pi\,x}{l}\right)$$

and ϕ_k is the solution of the IVP

$$\left(\frac{d^2}{dt^2}\phi_k(t)\right)+\left(\frac{a\,k\,\pi}{l}\right)^2\phi_k(t)=\alpha_k(t)$$
$$\phi_k(0)=0$$
$$\phi_k'(0)=0$$

where $\alpha_k(t)$ are Fourier coefficients of $f(t,x)$ with respect to the complete orthogonal system { $\sin\left(\frac{k\,\pi\,x}{l}\right)$ }

$$f(t,x)=\sum_{k=1}^{\infty}\alpha_k(t)\sin\left(\frac{k\,\pi\,x}{l}\right)$$

so, $\alpha_k(t)=\int_0^l f(t,x)\sin\left(\frac{k\,\pi\,x}{l}\right)dx$

Let's find the solution in the case of $a=1$, $l=\pi$, $f(t,x)=\cos(t)\sin(2x)$, $u_0(x)=\sin(x)-2\sin(3x)+\sin(4x)$, $u_1(x)=\sin(x)-\sin(2x)$

first we construct the orthogonal system { $\sin\left(\frac{k\,\pi\,x}{l}\right)$ }

```
> theta:=(k,x)->sin((k*Pi*x)/l);
```

$$\theta:=(k,x)\to\sin\left(\frac{k\,\pi\,x}{l}\right)$$

```
> theta(1,x);
```

$$\sin\left(\frac{\pi\,x}{l}\right)$$

```
> a:=k->2/l*int(u0(x)*sin(k*Pi*x/l),x = 0 .. l);
```

$$a:=k\to\frac{2}{l}\int_0^l \mathrm{u0}(x)\sin\left(\frac{k\,\pi\,x}{l}\right)dx$$

```
> b:=k->2/(A*k*Pi)*int(u1(x)*sin(k*Pi*x/l),x = 0 .. l);
```

$$b:=k\to\frac{2}{A\,k\,\pi}\int_0^l \mathrm{u1}(x)\sin\left(\frac{k\,\pi\,x}{l}\right)dx$$

We work with the partial sum which gives the solution for homogeneous equation, instead of infinity we put a big value (let's say 100)

```
> v:=(t,x)->
sum('(a(k)*cos(A*k*Pi*t/l)+b(k)*sin(A*k*Pi*t/l))*theta(k,x)','k' = 1 ..
100);
```

$$v := (t,x) \to \sum_{'k'=1}^{100} '\left(\mathrm{a}(k)\cos\left(\frac{A\,k\,\pi\,t}{l}\right) + \mathrm{b}(k)\sin\left(\frac{A\,k\,\pi\,t}{l}\right)\right)\theta(k,x)'$$

Now, we initialize the data:

```
> A:=1;l:=Pi;
```

$$A := 1$$

$$l := \pi$$

```
> f:=(t,x)->cos(t)*sin(2*x);
```

$$f := (t,x) \to \cos(t)\sin(2\,x)$$

```
> u0:=x->sin(x)-2*sin(3*x)+sin(4*x);
```

$$u0 := x \to \sin(x) - 2\sin(3\,x) + \sin(4\,x)$$

```
> u1:=x->sin(x)-sin(2*x);
```

$$u1 := x \to \sin(x) - \sin(2\,x)$$

```
> v(t,x);
```

$$(\cos(t)+\sin(t))\sin(x) - \frac{1}{2}\sin(2\,t)\sin(2\,x) - 2\cos(3\,t)\sin(3\,x) + \cos(4\,t)\sin(4\,x)$$

Now we construct the solution for the nonhomogeneous problem, w(t,x). First, we have to find the solution of the IVP's

```
>
> for k from 1 to 10 do
  alpha[k]:=unapply(int(f(t,x)*theta(k,x),x = 0 .. l),t);
end do;
```

$$\alpha_1 := 0$$

$$\alpha_2 := t \to \frac{1}{2}\pi\cos(t)$$

$$\alpha_3 := 0$$

$$\alpha_4 := 0$$

$$\alpha_5 := 0$$

$$\alpha_6 := 0$$

$$\alpha_7 := 0$$

$$\alpha_8 := 0$$

$$\alpha_9 := 0$$

$$\alpha_{10} := 0$$

```
> in_c:=phi(0)=0,D(phi)(0)=0;
```

$$in\_c := \phi(0) = 0, D(\phi)(0) = 0$$

```
> for k from 1 to 10 do
    d_eqs[k]:=diff(phi(t),t$2)+(A*k*Pi/l)^2*phi(t) = alpha[k](t);
    s:=dsolve({d_eqs[k],in_c},phi(t));
    phi_sol[k]:=unapply(rhs(s),t);
end do:
> w:=(t,x) -> sum('phi_sol[k](t)*theta(k,x)','k' = 1 .. 10);
```

$$w := (t, x) \to \sum_{'k' = 1}^{10} 'phi\_sol_k(t)\ \theta(k, x)'$$

```
> w(t,x);
```

$$\left(-\frac{1}{6}\cos(2\,t)\,\pi + \frac{1}{6}\pi\cos(t)\right)\sin(2\,x)$$

Therefore the solution of the problem is

```
> u:=(t,x)->v(t,x)+w(t,x);
```

$$u := (t, x) \to \mathrm{v}(t, x) + \mathrm{w}(t, x)$$

```
> u(t,x);
```

$$(\cos(t) + \sin(t))\sin(x) - \frac{1}{2}\sin(2\,t)\sin(2\,x) - 2\cos(3\,t)\sin(3\,x) + \cos(4\,t)\sin(4\,x) + \left(-\frac{1}{6}\cos(2\,t)\,\pi + \frac{1}{6}\pi\cos(t)\right)\sin(2\,x)$$

```
>
> plot3d(u(t,x),t=0..5,x=0..l);
```



```
> animate(u(t,x),x=0..l,t=0..5);
```




```
>
```